
BIE 5110/6110 
Sprinkle & Trickle Irrigation 

Fall Semester, 2004 
 
Assignment #7 (200 pts) 
Operating Point for Fixed Sprinkler System 
Due: 29 Oct 04 
 
Given: 
 

• A fixed sprinkler system (all sprinklers operate simultaneously) in an orchard. 
• Buried IPS-PVC 1-½” lateral pipes (see Table 8.3 of the textbook). 
• Rainbird® under-tree M20VH impact sprinklers with 5/64-inch SBN-1 nozzle (see 

www.rainbird.com for technical specifications). 
• Lateral spacing: Sl = 40.00 ft. 
• Sprinkler spacing: Se = 40.00 ft. 
• The field is trapezoidal in shape, as shown below: 

 

800 ft

1,100 ft

pump

m
ainline

560 ft

0.10%
uphill

0.18% downhill

 
 

• In the above figure, only some of the laterals and sprinklers are shown, but remember 
that the field area is covered with sprinklers in a fixed, permanent system. 

• There are 27 laterals along the mainline. 

• Note the field slopes along mainline and lateral directions. 
• The suction (upstream) side of the pump has a 4-ft static lift from a pond and 10 ft of 8” 

PVC pipe (ID = 8.205”) with one 90-degree elbow and a strainer screen at the inlet. 
• The mainline is 8” PVC pipe (ID = 8.205”), and is 1,100-ft long. 
• Sprinkler riser height is: hr = 3.0 ft. 
• A Berkeley model 4GQH pump curve as shown on the following page.  You will use 

the characteristic curve for 1600 RPM. 
• Ignore minor losses along the mainline and laterals. 

 
Required: 
 

1. Determine the irrigated area (acres). 
2. Develop an equation for sprinkler flow rate (q) as a function of pressure (P), whereby q 

= KdPx (you determine Kd and x based on manufacturer’s data).  Note that for a 
straight-bore nozzle, you would expect x to be very close to 0.5, as in Eq. 5.1a. 

3. Determine the number of sprinklers along each lateral, where lateral #1 is the closest 
to the pump, and lateral #27 is the furthest from the pump. 

4. Develop at least five points on the system curve and present those points numerically 
in a table.  You should write a macro (Excel) or computer program to do this (don’t 
attempt to do it by hand with only a calculator). 

5. Plot the system curve points on the attached pump curve graph, and draw a smooth 
curve through the points. 

6. Determine the operating point (Q in gpm & TDH in ft) for this system. 
7. Determine the average application rate for the whole field area. 
 

Do you work in an organized, neat way.  Make comments about assumptions 
and other technical issues as appropriate.  Turn in all your work, including the 

code for your macro or computer program. 



Solution: 
 

1. The irrigated area is approximately: 
 

0.5(800 560)(1100)A 17.2 acres
43,560
+

= =  

 
Due to the trapezoidal field shape, and the fact that there are 27 laterals at Sl = 40 ft, 
the effective irrigated area is slightly less than 17.2 acres. 
 

2. A linear regression is performed on logarithms of the manufacturer’s data for P and q 
(Rainbird® M20VH with 5/64-inch SBN-1 nozzle): 
 

Pressure 
(psi)

Flow 
(gpm) ln(P) ln(q)

25 0.88 3.2189 -0.1278
30 0.97 3.4012 -0.0305
35 1.05 3.5553 0.0488
40 1.12 3.6889 0.1133
45 1.19 3.8067 0.1740
50 1.25 3.9120 0.2231  

 
given an R2 of 0.9996, and the following equation: 
 

0.506q 0.173P=  
 
for q in gpm; and P in psi.  Note that the exponent is close to 0.500, which is expected 
for a straight-bore nozzle.  But notice also that allowing for the flexibility in the 
exponent, x, gives a better mathematical fit to the manufacturer’s data. 
 

3. Let the first lateral be located at ½Sl from the lower edge of the field (where the pump 
is located).  Calculate the number of sprinklers per lateral by rounding the potential 
lateral length by Se.  The potential lateral length is calculated by linear interpolation 
along the left side of the field area (see the figure given above).  The equation is given 
below, and the calculation results are shown in the following table. 

 
800 560L 800 (1,100 y)

1,100
−⎛ ⎞= − − ⎜ ⎟

⎝ ⎠
 

 
where y is the distance along the mainline (ft); and L is the potential lateral length (ft). 

Lateral
Distance 

(ft)
Length 

(ft)
No. of 

Sprinklers
1 20             564.4 14
2 60             573.1 14
3 100           581.8 15
4 140           590.5 15
5 180           599.3 15
6 220           608.0 15
7 260           616.7 15
8 300           625.5 16
9 340           634.2 16

10 380           642.9 16
11 420           651.6 16
12 460           660.4 17
13 500           669.1 17
14 540           677.8 17
15 580           686.5 17
16 620           695.3 17
17 660           704.0 18
18 700           712.7 18
19 740           721.5 18
20 780           730.2 18
21 820           738.9 18
22 860           747.6 19
23 900           756.4 19
24 940           765.1 19
25 980           773.8 19
26 1,020        782.5 20
27 1,060        791.3 20  

 
4. Develop a computer program to start with a given pressure at the furthest downstream 

sprinkler on lateral #27, calculate the flow rate at that sprinkler, calculate the pressure 
at the next upstream sprinkler, then the flow rate at that sprinkler, and so on, until 
reaching the mainline.  Calculate the pressure in the mainline at the location of lateral 
#26, then iterate along lateral #26 to get the same pressure in the mainline at that 
location.  Repeat for all other laterals, moving in the upstream direction, until a 
pressure is obtained for the upstream end of the mainline.  The system flow rate is 
known from these calculations (sum of all individual sprinklers). Assume pipe leakage 
is zero. 
 
Knowing the system flow rate, determine the losses in the suction side of the pipe, and 
determine TDH by adding the velocity head at the beginning of the mainline, the 
pressure head at the beginning of the mainline, the static lift on the suction side, and 
the hydraulic losses on the suction side of the pump. 

Preliminary calculations and assumptions: 
 
Assume a water temperature of 10°C, giving a kinematic viscosity of: 1.306(10)-6 m2/s, 
or 1.406(10)-5 ft2/s. 
 
Use the Swamee-Jain equation with ε = 1.5(10)-6 m, or 4.92(10)-6 ft for PVC to obtain 
the Darcy-Weisbach friction factor, f. 
 
From Table 8.3, the ID of the lateral pipe is 1.754 inches (0.1462 ft).  The ID of the 
mainline pipe is 8.205 inches (0.6838 ft). 
 
Once the pressure at the upstream end of the mainline is calculated, the additional 
TDH values include static lift, velocity head, and losses in the suction side of the pump, 
plus the riser height.  Assume the pump outlet is at the same elevation as the 
upstream end of the mainline (you don’t know if the mainline or laterals are buried, nor 
how deep they might be, but this information could be used to develop a somewhat 
more specific design). 
 

( )
2

main lift r f suction

VTDH 2.308P h h h
2g

= + + + +  

 
where TDH is in ft; Pmain is the pressure at the upstream end of the mainline (psi); hlift is 
given as 4.0 ft; hr is given as 3.0 ft; (hf)suction are the hydraulic losses in the suction pipe 
(ft); and the last term is the velocity head (ft). 
 
The iterative part is to determine Pmain; the rest of the TDH terms are easy to calculate 
directly. 
 
From Table 11.2 (minor loss coefficients): 
 

8-inch “basket strainer”: Kr = 0.75 
8-inch “regular 90-deg elbow”: Kr = 0.26 

 
Velocity head in suction pipe: 
 

2 2
2

2 4

V 8Q 0.1151Q
2g g D

= =
π

 

 
Friction loss in suction pipe: 
 

2
2

f
L Vh f 1.684 f Q
D 2g

= =  

Putting it all together: 
 

( )2 2
mainTDH 2.308P 4.0 3.0 1.684 f Q 0.1151Q 1 0.75 0.26= + + + + + +  

 
or, 

[ ]2
mainTDH 2.308P 7.0 Q 1.684 f 0.2314= + + +  

 
The results are given in the table below: 
 

P (psi)
Qs 

(gpm)
Pmain 

(psi) Re f
TDH 
(ft)

20 367.2 21.8 108,347 0.01761 57.50
25 411.2 27.2 121,318 0.01721 70.09
30 451.0 32.7 133,061 0.01689 82.66
35 487.6 38.1 143,853 0.01663 95.21
40 521.6 43.5 153,902 0.01641 107.74
45 553.6 48.9 163,344 0.01622 120.26
50 583.9 54.3 172,277 0.01605 132.77
55 612.7 59.7 180,777 0.01590 145.28
60 640.2 65.1 188,901 0.01577 157.77  

 
where P is the pressure at the furthest downstream sprinkler on lateral #27; Qs is the 
total system flow rate; Pmain is the pressure at the upstream end of the mainline (just 
downstream of the pump); Re is the Reynold’s number in the suction pipe; f is the value 
from Swamee-Jain; and TDH is the total dynamic head. 
 

5. The system curve (Qs versus TDH) is superimposed upon the pump manufacturer’s 
curves, as shown in the figure below. 
 

6. The operating point (intersection of the system curve and the 1600 RPM pump curve) 
is seen to be approximately: 
 

Qs = 568 gpm 
TDH = 126 ft 

 
7. The average application rate at this operating point is approximately: 

 

avg
(568)(12)(3600)AR 0.075 inch/hr

(458)(40)(40)(448.86)
= =  

 
or 1.9 mm/hr. 

The MS Excel VBA listing follows: 
 
Const Pi = 3.141593 
Const Se = 40#          'ft 
Const Sl = 40#          'ft 
Const LatCount = 27     'number of laterals 
Const Dlat = 0.1462     'ft 
Const Dmain = 0.6838    'ft 
Const SoLat = -0.0018   'ft/ft 
Const SoMain = 0.001    'ft/ft 
 
Function hf(ByVal Q As Double, ByVal D As Double, ByVal L As Double) As Double 
'------------------------------------------------------------------------------ 
' Returns friction loss in feet of water head (Darcy-Weisbach). 
' If turbulent, uses the Swamee-Jain equation for the friction factor, f. 
'------------------------------------------------------------------------------ 
 
  Dim Re As Double, RelRough As Double, f As Double 
     
  viscosity = 0.00001406  'ft2/s 
   
  RelRough = 0.00000492 / D 
  Re = 4 * Q / (viscosity * Pi * D) 
   
  If Re > 4000 Then 
    f = RelRough / 3.75 + 5.74 / Re ^ 0.9   'Turbulent 
    f = WorksheetFunction.Log10(f) 
    f = 0.25 / f ^ 2 
  Else 
    f = 64 / Re                             'Laminar 
  End If 

  hf = 8# * f * L * (Q / Pi) ^ 2 / (32.2 * D ^ 5) 
   
End Function 
 
Function Lateral(ByVal P As Double, n As Integer, Qlat As Double) As Double 
'------------------------------------------------------------------------------ 
' Calculates lateral inlet pressure head for "n" sprinklers. 
' Pressure, P, is in psi. 
'------------------------------------------------------------------------------ 
   
  Dim Qs As Double, h As Double 
   
  Qlat = 0# 'cfs 
   
  For i = 1 To n 
    Qs = 0.173 * P ^ 0.506                              'gpm 
    Qlat = Qlat + Qs / 448.86                           'cfs 
    h = P * 2.308 + hf(Qlat, Dlat, Se) + SoLat * Se     'ft 
    P = h / 2.308                                       'psi 
  Next 
   
  Lateral = h 
   
End Function 
 
Function ParabolaFit(x, y, target As Double) As Double 
'------------------------------------------------------------------------------ 
' Determines constants a, b, and c for a parabola through three points. 
' Equation is: y = ax^2 + bx + c.  If parabola impossible, uses bisection. 
' Returns the x-value (P) which matches the specified target y-value (hmain). 
' P is the pressure at the furthest DS sprinkler in the lateral. 
'------------------------------------------------------------------------------ 
 
  Dim Bisection As Boolean 
  Dim foo As Double, boo As Double 
  Dim a As Double, b As Double, c As Double 
   
  ParabolaFit = 0 
  foo = x(2) - x(1) 
  Bisection = Abs(foo) < 0.0000000001 
   
  If Not Bisection Then 
   
    C0 = x(1) ^ 2 
    c1 = (x(3) - x(1)) / foo 
    c2 = x(2) ^ 2 - C0 
     
    boo = x(3) ^ 2 - C0 - c1 * c2 
    Bisection = Abs(boo) < 0.0000000001 
     
    If Not Bisection Then 
   
      '------------------------ 
      ' Parabolic interpolation 
      '------------------------ 
     
      a = ((y(1) - y(2)) * c1 - y(1) + y(3)) / boo 
      b = (y(2) - y(1) - a * c2) / foo 
      c = y(2) - x(2) * (a * x(2) + b) 
       
      c = c - target 
      ParabolaFit = (-b + Sqr(b * b - 4 * a * c)) / (2 * a) 
      Exit Function 
    End If 
  End If 

  If Bisection Then 
   
    '-------------- 
    ' Use bisection 
    '-------------- 
     
    If y(2) < target Then 
      ParabolaFit = (y(2) + y(3)) / 2 
    Else 
      ParabolaFit = (y(2) + y(1)) / 2 
    End If 
  End If 
     
End Function 
 
Function QsPmain(ByVal P As Double, Flow As Boolean) As Double 
'------------------------------------------------------------------------------ 
' Iterates to determine the system flow rate for a given starting pressure. 
' Returns either flow rate (Qs) or pressure (Pmain) at US end of the mainline. 
'------------------------------------------------------------------------------ 
 
  Dim n As Integer 
  Dim lat As Integer 
  Dim LatHead(1 To 3) As Double 
  Dim Pressure(1 To 3) As Double 
  Dim Qlat As Double, x As Double 
  Dim Qmain As Double, L As Double, NewP As Double, hmain As Double 
     
  hmain = 0 
  Qmain = 0 
   
  For lat = LatCount To 1 Step -1 
   
    '-------------------------------------------- 
    ' Determine number of sprinklers this lateral 
    '-------------------------------------------- 
     
    x = 20 + (lat - 1) * Sl 
    L = 800# - (1100# - x) * 0.2181818 
    n = Round(L / Se, 0) 
     
    '-------------------------------------- 
    ' Calculate lateral inlet pressure head 
    '-------------------------------------- 
     
    If lat = LatCount Then 
     
      '------------------------------------ 
      ' No need to iterate for last lateral 
      '------------------------------------ 
       
      LatHead(2) = Lateral(P, n, Qlat) 
    Else 
     
      '------------------------------------------------ 
      ' Iterate to match lateral inlet & mainline heads 
      '------------------------------------------------ 
       
      Pressure(1) = P / 4 
      Pressure(3) = 3 * P 
      Pressure(2) = (Pressure(1) + Pressure(3)) / 2 
       
      LatHead(1) = Lateral(Pressure(1), n, Qlat) 
      LatHead(3) = Lateral(Pressure(3), n, Qlat) 
      LatHead(2) = Lateral(Pressure(2), n, Qlat) 

      If (LatHead(1) > hmain) Or (LatHead(3) < hmain) Then 
 
        '------------------------------- 
        ' Failed to bracket the solution 
        '------------------------------- 
   
        QsPmain = -100 
        Exit Function 
      End If 
       
      For i = 1 To 50 
       
        '---------------------------------- 
        ' Search by parabolic interpolation 
        '---------------------------------- 
 
        NewP = ParabolaFit(Pressure, LatHead, hmain) 
         
        If NewP < Pressure(2) Then 
          Pressure(3) = Pressure(2) 
          LatHead(3) = LatHead(2) 
        Else 
          Pressure(1) = Pressure(2) 
          LatHead(1) = LatHead(2) 
        End If 
         
        Pressure(2) = NewP 
        LatHead(2) = Lateral(NewP, n, Qlat) 
         
        If Abs(LatHead(2) - hmain) < 0.001 Then 
         
          '------------------- 
          ' Solution converged 
          '------------------- 
           
          Exit For 
        End If 
      Next 
       
    End If 
     
    '----------------------------------------- 
    ' Move upstream one hydrant along mainline 
    '----------------------------------------- 
     
    Qmain = Qmain + Qlat 
    hmain = LatHead(2) + hf(Qmain, Dmain, Sl) + SoMain * Sl 
  Next 
 
  '---------------------------- 
  ' Return the system flow rate 
  ' or the mainline pressure 
  '---------------------------- 
   
  If Flow Then 
    QsPmain = Qmain * 448.86  'gpm 
  Else 
    QsPmain = hmain / 2.308   'psi 
  End If 
   
End Function 


